QT       += core gui

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets

CONFIG += c++17

HEADERS += \
    gpu.h \
    gpu_global.h \
    lineraster.h \
    triangleraster.h \
    ../../Public/mathFunctions.h \
    ../../Public/mymath.h \
    ../../Public/myvector.h \
    ../../Public/matrix.h \
    ../../Public/datatype.h

SOURCES += \
    gpu.cpp \
    lineraster.cpp \
    triangleraster.cpp

DEFINES += GPU_LIB

TEMPLATE = lib

#包含目录
INCLUDEPATH +=  $$PWD/../../Public \
                $$PWD/../../Plugins/Log \
                $$PWD/../../Thirdpart/log4cplus-121/include
#导入导出库

 contains(QT_ARCH, x86_64) {

    CONFIG(debug, debug|release):{ #_DEBUG_X64


         DESTDIR = $$PWD/../../../bin/x64/debug

         LIBS += -L$$DESTDIR -lLogD

         TARGET = GpuD

     }
    else:CONFIG(release, debug|release):{#_RELEASE_X64

         DESTDIR = $$PWD/../../../bin/x64/release

         LIBS += -L$$DESTDIR -lLog

         TARGET = Gpu

     }
}
else {

    CONFIG(debug, debug|release):{ #_DEBUG_X86

         DESTDIR = $$PWD/../../../bin/x86/debug

         LIBS += -L$$DESTDIR -lLogD

         TARGET = GpuD

     }
    else:CONFIG(release, debug|release):{#_RELEASE_X86

         DESTDIR = $$PWD/../../../bin/x86/release

         LIBS += -L$$DESTDIR -lLog

         TARGET = Gpu

     }

}
